NAG Library Routine Document

S30NAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S30NAF computes the European option price given by Heston's stochastic volatility model.

2 Specification

```
SUBROUTINE S30NAF (CALPUT, M, N, X, S, T, SIGMAV, KAPPA, CORR, VAR0,          &
                   ETA, GRISK, R, Q, P, LDP, IFAIL)
INTEGER            M, N, LDP, IFAIL
REAL (KIND=nag_wp) X(M), S, T(N), SIGMAV, KAPPA, CORR, VAR0, ETA,             &
                   GRISK, R, Q, P(LDP,N)
CHARACTER(1)       CALPUT
```

3 Description

S30NAF computes the price of a European option using Heston's stochastic volatility model. The return on the asset price, S, is

$$\frac{dS}{S} = (r-q)dt + \sqrt{v_t}dW_t^{(1)}$$

and the instantaneous variance, v_t, is defined by a mean-reverting square root stochastic process,

$$dv_t = \kappa(\eta - v_t)dt + \sigma_v\sqrt{v_t}dW_t^{(2)},$$

where r is the risk free annual interest rate; q is the annual dividend rate; v_t is the variance of the asset price; σ_v is the volatility of the volatility, $\sqrt{v_t}$; κ is the mean reversion rate; η is the long term variance. $dW_t^{(i)}$, for $i = 1, 2$, denotes two correlated standard Brownian motions with

$$\mathbb{C}\mathrm{ov}\left[dW_t^{(1)}, dW_t^{(2)}\right] = \rho dt.$$

The option price is computed by evaluating the integral transform given by Lewis (2000) using the form of the characteristic function discussed by Albrecher *et al.* (2007), see also Kilin (2006).

$$P_{\text{call}} = Se^{-qT} - Xe^{-rT}\frac{1}{\pi}\mathrm{Re}\left[\int_{0+i/2}^{\infty+i/2} e^{-ik\bar{X}}\frac{\hat{H}(k,v,T)}{k^2-ik}\mathrm{d}k\right], \tag{1}$$

where $\bar{X} = \ln(S/X) + (r-q)T$ and

$$\hat{H}(k,v,T) = \exp\left(\frac{2\kappa\eta}{\sigma_v^2}\left[t\mathrm{gendgroup} - \ln\left(\frac{1-\mathrm{he}^{-\xi t}}{1-\mathrm{h}}\right)\right] + \mathrm{v_t g}\left[\frac{1-\mathrm{e}^{-\xi t}}{1-\mathrm{he}^{-\xi t}}\right]\right),$$

$$g = \frac{1}{2}(b-\xi), \quad h = \frac{b-\xi}{b+\xi}, \quad t = \sigma_v^2 T/2,$$

$$\xi = \left[b^2 + 4\frac{k^2-ik}{\sigma_v^2}\right]^{\frac{1}{2}},$$

$$b = \frac{2}{\sigma_v^2}\left[(1 - \gamma + ik)\rho\sigma_v + \sqrt{\kappa^2 - \gamma(1 - \gamma)\sigma_v^2}\right]$$

with $t = \sigma_v^2 T/2$. Here γ is the risk aversion parameter of the representative agent with $0 \leq \gamma \leq 1$ and $\gamma(1 - \gamma)\sigma_v^2 \leq \kappa^2$. The value $\gamma = 1$ corresponds to $\lambda = 0$, where λ is the market price of risk in Heston (1993) (see Lewis (2000) and Rouah and Vainberg (2007)).

The price of a put option is obtained by put-call parity.

The option price $P_{ij} = P(X = X_i, T = T_j)$ is computed for each strike price in a set X_i, $i = 1, 2, \ldots, m$, and for each expiry time in a set T_j, $j = 1, 2, \ldots, n$.

4 References

Albrecher H, Mayer P, Schoutens W and Tistaert J (2007) The little Heston trap *Wilmott Magazine* **January 2007** 83–92

Heston S (1993) A closed-form solution for options with stochastic volatility with applications to bond and currency options *Review of Financial Studies* **6** 327–343

Kilin F (2006) Accelerating the calibration of stochastic volatility models *MPRA Paper No. 2975* http://mpra.ub.uni-muenchen.de/2975/

Lewis A L (2000) Option valuation under stochastic volatility *Finance Press, USA*

Rouah F D and Vainberg G (2007) *Option Pricing Models and Volatility using Excel-VBA* John Wiley and Sons, Inc

5 Parameters

1: CALPUT – CHARACTER(1) *Input*

On entry: determines whether the option is a call or a put.

CALPUT = 'C'
A call; the holder has a right to buy.

CALPUT = 'P'
A put; the holder has a right to sell.

Constraint: CALPUT = 'C' or 'P'.

2: M – INTEGER *Input*

On entry: the number of strike prices to be used.

Constraint: M $\geq$ 1.

3: N – INTEGER *Input*

On entry: the number of times to expiry to be used.

Constraint: N $\geq$ 1.

4: X(M) – REAL (KIND=nag_wp) array *Input*

On entry: X(i) must contain X_i, the ith strike price, for $i = 1, 2, \ldots,$ M.

Constraint: X(i) $\geq z$ and X(i) $\leq 1/z$, where $z =$ X02AMF(), the safe range parameter, for $i = 1, 2, \ldots,$ M.

5: S – REAL (KIND=nag_wp) *Input*

On entry: S, the price of the underlying asset.

Constraint: $\text{S} \geq z$ and $\text{S} \leq 1.0/z$, where $z = \text{X02AMF}()$, the safe range parameter.

6: T(N) – REAL (KIND=nag_wp) array *Input*

On entry: $\text{T}(i)$ must contain T_i, the ith time, in years, to expiry, for $i = 1, 2, \ldots, \text{N}$.

Constraint: $\text{T}(i) \geq z$, where $z = \text{X02AMF}()$, the safe range parameter, for $i = 1, 2, \ldots, \text{N}$.

7: SIGMAV – REAL (KIND=nag_wp) *Input*

On entry: the volatility, σ_v, of the volatility process, $\sqrt{v_t}$. Note that a rate of 20% should be entered as 0.2.

Constraint: $\text{SIGMAV} > 0.0$.

8: KAPPA – REAL (KIND=nag_wp) *Input*

On entry: κ, the long term mean reversion rate of the volatility.

Constraint: $\text{KAPPA} > 0.0$.

9: CORR – REAL (KIND=nag_wp) *Input*

On entry: the correlation between the two standard Brownian motions for the asset price and the volatility.

Constraint: $-1.0 \leq \text{CORR} \leq 1.0$.

10: VAR0 – REAL (KIND=nag_wp) *Input*

On entry: the initial value of the variance, v_t, of the asset price.

Constraint: $\text{VAR0} \geq 0.0$.

11: ETA – REAL (KIND=nag_wp) *Input*

On entry: η, the long term mean of the variance of the asset price.

Constraint: $\text{ETA} > 0.0$.

12: GRISK – REAL (KIND=nag_wp) *Input*

On entry: the risk aversion parameter, γ, of the representative agent.

Constraint: $0.0 \leq \text{GRISK} \leq 1.0$ and $\text{GRISK} \times (1.0 - \text{GRISK}) \times \text{SIGMAV} \times \text{SIGMAV} \leq \text{KAPPA} \times \text{KAPPA}$.

13: R – REAL (KIND=nag_wp) *Input*

On entry: r, the annual risk-free interest rate, continuously compounded. Note that a rate of 5% should be entered as 0.05.

Constraint: $\text{R} \geq 0.0$.

14: Q – REAL (KIND=nag_wp) *Input*

On entry: q, the annual continuous yield rate. Note that a rate of 8% should be entered as 0.08.

Constraint: $\text{Q} \geq 0.0$.

15: P(LDP, N) – REAL (KIND=nag_wp) array *Output*

On exit: $\text{P}(i,j)$ contains P_{ij}, the option price evaluated for the strike price X_i at expiry T_j for $i = 1, 2, \ldots, \text{M}$ and $j = 1, 2, \ldots, \text{N}$.

16: LDP – INTEGER *Input*

On entry: the first dimension of the array P as declared in the (sub)program from which S30NAF is called.

Constraint: LDP $\geq$ M.

17: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, CALPUT $= \langle$*value*$\rangle$ was an illegal value.

IFAIL $= 2$

On entry, M $= \langle$*value*$\rangle$.
Constraint: M ≥ 1.

IFAIL $= 3$

On entry, N $= \langle$*value*$\rangle$.
Constraint: N ≥ 1.

IFAIL $= 4$

On entry, X($\langle$*value*$\rangle$) $= \langle$*value*$\rangle$.
Constraint: X(i) $\geq \langle$*value*$\rangle$ and X(i) $\leq \langle$*value*$\rangle$.

IFAIL $= 5$

On entry, S $= \langle$*value*$\rangle$.
Constraint: S $\geq \langle$*value*$\rangle$ and S $\leq \langle$*value*$\rangle$.

IFAIL $= 6$

On entry, T($\langle$*value*$\rangle$) $= \langle$*value*$\rangle$.
Constraint: T(i) $\geq \langle$*value*$\rangle$.

IFAIL $= 7$

On entry, SIGMAV $= \langle$*value*$\rangle$.
Constraint: SIGMAV > 0.0.

IFAIL $= 8$

On entry, KAPPA $= \langle value \rangle$.
Constraint: KAPPA > 0.0.

IFAIL $= 9$

On entry, CORR $= \langle value \rangle$.
Constraint: $|\text{CORR}| \leq 1.0$.

IFAIL $= 10$

On entry, VAR0 $= \langle value \rangle$.
Constraint: VAR0 ≥ 0.0.

IFAIL $= 11$

On entry, ETA $= \langle value \rangle$.
Constraint: ETA > 0.0.

IFAIL $= 12$

On entry, GRISK $= \langle value \rangle$, SIGMAV $= \langle value \rangle$ and KAPPA $= \langle value \rangle$.
Constraint: $0.0 \leq \text{GRISK} \leq 1.0$ and $\text{GRISK} \times (1.0 - \text{GRISK}) \times \text{SIGMAV}^2 \leq \text{KAPPA}^2$.

IFAIL $= 13$

On entry, R $= \langle value \rangle$.
Constraint: R ≥ 0.0.

IFAIL $= 14$

On entry, Q $= \langle value \rangle$.
Constraint: Q ≥ 0.0.

IFAIL $= 16$

On entry, LDP $= \langle value \rangle$ and M $= \langle value \rangle$.
Constraint: LDP $\geq$ M.

IFAIL $= 17$

Quadrature has not converged to the specified accuracy. However, the result should be a reasonable approximation.

IFAIL $= 18$

Solution cannot be computed accurately. Check values of input parameters.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

The accuracy of the output is determined by the accuracy of the numerical quadrature used to evaluate the integral in (1). An adaptive method is used which evaluates the integral to within a tolerance of $\max\left(10^{-8}, 10^{-10} \times |I|\right)$, where $|I|$ is the absolute value of the integral.

8 Parallelism and Performance

S30NAF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example computes the price of a European call using Heston's stochastic volatility model. The time to expiry is 6 months, the stock price is 100 and the strike price is 100. The risk-free interest rate is 5% per year, the volatility of the variance, σ_v, is 22.5% per year, the mean reversion parameter, κ, is 2.0, the long term mean of the variance, η, is 0.01 and the correlation between the volatility process and the stock price process, ρ, is 0.0. The risk aversion parameter, γ, is 1.0 and the initial value of the variance, VAR0, is 0.01.

10.1 Program Text

```
    Program s30nafe

!     S30NAF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, s30naf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: corr, eta, grisk, kappa, q, r, s,    &
                                          sigmav, var0
      Integer                          :: i, ifail, j, ldp, m, n
      Character (1)                    :: calput
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: p(:,:), t(:), x(:)
!     .. Executable Statements ..
      Write (nout,*) 'S30NAF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) calput
      Read (nin,*) s, r, q
      Read (nin,*) kappa, eta, var0, sigmav, corr, grisk
      Read (nin,*) m, n

      ldp = m
      Allocate (p(ldp,n),t(n),x(m))

      Read (nin,*)(x(i),i=1,m)
```

```
      Read (nin,*)(t(i),i=1,n)

      ifail = 0
      Call s30naf(calput,m,n,x,s,t,sigmav,kappa,corr,var0,eta,grisk,r,q,p,ldp, &
        ifail)

      Write (nout,*)
      Write (nout,*) 'Heston''s Stochastic volatility Model'

      Select Case (calput)
      Case ('C','c')
        Write (nout,*) 'European Call :'
      Case ('P','p')
        Write (nout,*) 'European Put :'
      End Select

      Write (nout,99998) '  Spot                  = ', s
      Write (nout,99998) '  Volatility of vol     = ', sigmav
      Write (nout,99998) '  Mean reversion        = ', kappa
      Write (nout,99998) '  Correlation           = ', corr
      Write (nout,99998) '  Variance              = ', var0
      Write (nout,99998) '  Mean of variance      = ', eta
      Write (nout,99998) '  Risk aversion         = ', grisk
      Write (nout,99998) '  Rate                  = ', r
      Write (nout,99998) '  Dividend              = ', q

      Write (nout,*)
      Write (nout,*) '   Strike    Expiry    Option Price'

      Do i = 1, m

        Do j = 1, n
          Write (nout,99999) x(i), t(j), p(i,j)
        End Do

      End Do

99999 Format (1X,2(F9.4,1X),6X,F9.4)
99998 Format (A,1X,F8.4)
    End Program s30nafe
```

10.2 Program Data

```
S30NAF Example Program Data
 'C'                          : Call = 'C', Put = 'P'
 100.0 0.05 0.0               : S, R, Q
 2.0 0.01 0.01 0.225 0.0 1.0 : KAPPA, ETA, VAR0, SIGMAV, CORR, GRISK
 1 1                          : M, N
 100.0                        : X(I), I = 1,2,...N
 0.5                          : T(I), I = 1,2,...M
```

10.3 Program Results

```
 S30NAF Example Program Results

 Heston's Stochastic volatility Model
 European Call :
  Spot                  =  100.0000
  Volatility of vol     =    0.2250
  Mean reversion        =    2.0000
  Correlation           =    0.0000
  Variance              =    0.0100
  Mean of variance      =    0.0100
  Risk aversion         =    1.0000
```

```
 Rate                   =    0.0500
 Dividend               =    0.0000

   Strike    Expiry    Option Price
 100.0000    0.5000          4.0851
```